Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 2
Symbolic Computation

Initializations

```
> restart;
>
```

2.1 General Symbolic Computation

The real power of Maple lies in its symbolic capabilities. A computer algebra system allows the user to perform mathematical computations on the computer screen much like they used to be performed with pencil and paper. We now explore some of these symbolic features.

Examples

Example 2.1.1

Differentiation of expressions.

To differentiate a mathematical expression, one uses the **diff** command. We define the expression $x^2 \sin 7x$ and compute its first and second derivative.

```
> expr:=x^2*sin(7*x);
```

$$expr := x^2 \sin(7x) \tag{2.1.1.1}$$

```
> der1:=diff(expr, x);
```

$$der1 := 2x \sin(7x) + 7x^2 \cos(7x) \tag{2.1.1.2}$$

```
> der2:=diff(expr, x$2);
```

$$der2 := 2\sin(7x) + 28x\cos(7x) - 49x^2 \sin(7x) \tag{2.1.1.3}$$

```
>
```

Example 2.1.2

Integration of expressions.

To integrate a mathematical expression, one uses the **int** command. We compute

$$\int \sqrt{x^2+7}\, dx$$

```
> expr:=sqrt(x^2+7);
```

$$expr := \sqrt{x^2+7} \tag{2.1.2.1}$$

```
> e1:=int(expr, x)+C;
```

$$e1 := \frac{1}{2} x \sqrt{x^2+7} + \frac{7}{2} \operatorname{arcsinh}\left(\frac{1}{7}\sqrt{7}\, x\right) + C \tag{2.1.2.2}$$

Alternatively, we could first have "pretty printed" the integral using the inert version **Int** of

the **int** command.

```
> e1:=Int(sqrt(x^2+7), x);
```

$$e1 := \int \sqrt{x^2+7}\, dx \tag{2.1.2.3}$$

This integral can be evaluated using the **value** command.

```
> e2:=value(e1)+C;
```

$$e2 := \frac{1}{2} x \sqrt{x^2+7} + \frac{7}{2} \operatorname{arcsinh}\left(\frac{1}{7} \sqrt{7}\, x \right) + C \tag{2.1.2.4}$$

If you do not care for inverse hyperbolic functions, then you can easily convert this expression to logarithms.

```
> e3:=convert(e2, ln);
```

$$e3 := \frac{1}{2} x \sqrt{x^2+7} + \frac{7}{2} \ln\left(\frac{1}{7} \sqrt{7}\, x + \frac{1}{7} \sqrt{7 x^2+49} \right) + C \tag{2.1.2.5}$$

Maple contains a large variety of simplification routines. Knowing which to use when requires experience and familiarity with the system. In most cases the general purpose **simplify** command will do a good job.

```
> e4:=simplify(e3);
```

$$e4 := \frac{1}{2} x \sqrt{x^2+7} - \frac{7}{4} \ln(7) + \frac{7}{2} \ln\left(x + \sqrt{x^2+7} \right) + C \tag{2.1.2.6}$$

```
>
```

Example 2.1.3

Equation solving.

We consider two very familiar problems, the quadratic equation, and a linear system in two unknowns. The code will speak for itself.

```
> eq1:=a*x^2+b*x+c=0;
```

$$eq1 := a x^2 + b x + c = 0 \tag{2.1.3.1}$$

```
> sol1:=solve(eq1, x);
```

$$sol1 := -\frac{1}{2} \frac{b - \sqrt{b^2 - 4 a c}}{a}, -\frac{1}{2} \frac{b + \sqrt{b^2 - 4 a c}}{a} \tag{2.1.3.2}$$

```
>
```

A system of equations is entered as a Maple set.

```
> eq2:={3*x-4*y=1, 5+x+7*y=-3};
```

$$eq2 := \{3x - 4y = 1, 5 + x + 7y = -3\} \tag{2.1.3.3}$$

```
> sol2:=solve(eq2, {x, y});
```

$$sol2 := \{x = -1, y = -1\} \tag{2.1.3.4}$$

```
>
```

Most real life equations cannot be solved in closed form, but require numerical routines in stead. Maple's numeric equation solver is called **fsolve**. As an example we compute the points of intersection of the curves

$$y = 5x \quad \text{and} \quad y = e^x$$

First, we define the curves and make a sketch.

```
> c1:=5*x;
  c2:=exp(x);
```

$$c1 := 5\,x$$

$$c2 := e^{x} \qquad (2.1.3.5)$$

```
> plot({c1, c2}, x=0..3);
```



Clearly, there is a point of intersection on the interval [0, 0.5] and another on the interval [2, 3]. The fsolve routine quickly finds the *x*-coordinates of the points of intersection

```
> x_1:=fsolve(c1=c2, x=0..0.5);
```

$$x\_1 := 0.2591711018 \qquad (2.1.3.6)$$

```
> x_2:=fsolve(c1=c2, x=2..3);
```

$$x\_2 := 2.542641358 \qquad (2.1.3.7)$$

```
>
```

The **subs** (substitution) command can be applied to obtain the corresponding *y*-coordinates.

```
> P_1:=subs(x=x_1, [x, c1]);
```

$$P\_1 := [0.2591711018, 1.295855509] \qquad (2.1.3.8)$$

```
> P_2:=subs(x=x_2, [x, c1]);
```

$$P\_2 := [2.542641358, 12.71320679] \qquad (2.1.3.9)$$

```
>
```